

TRAVAIL D'ÉTUDES ET DE RECHERCHE

---

ASSISTANT DE PREUVES ET  
FORMALISATION

---

ENKI SOUILLOT

ENCADRÉ PAR VINCENT BEFFARA

# Table des matières

1	INTRODUCTION . . . . .	2
2	LEAN ET SES TACTIQUES . . . . .	3
	2.1. LES TACTIQUES DE RÉOLUTION — EXEMPLE . . . . .	3
	2.2. LES TACTIQUES DE TRAVAIL . . . . .	6
3	L'ANALYSE COMPLEXE PAR LEAN . . . . .	6
	3.1. CADRE . . . . .	6
	3.2. CAUCHY-RIEMANN, ÉTAPE 1 — ÉNONCÉ . . . . .	7
	3.3. UN PREMIER ESSAI . . . . .	8
	3.4. UN SIMPLE CHANGEMENT . . . . .	11
	3.5. CAUCHY-RIEMANN STEP 2 . . . . .	14

## 1 Introduction

Ce rapport a pour but de donner les bases de l'utilisation du langage de formalisation mathématique  $\text{LEAN}$ , à l'aide d'exemples plus ou moins simples de niveau Master.

Mais tout d'abord, qu'est-ce que  $\text{LEAN}$ ?  $\text{LEAN}$  (ou plutôt le " $\text{LEAN}$  Theorem Prover") est un langage de formalisation des théorèmes et preuves mathématiques développé par *Leonardo de Moura* au sein de *Microsoft Research*.

L'objectif de  $\text{LEAN}$  est la vérification des preuves mathématiques, notamment via l'application pure des règles de logique élémentaires.

Mais pourquoi vérifier informatiquement nos preuves, même lorsque celles-ci nous semblent justes de part en part? Le raisonnement mathématique repose sur la rigueur et la logique, il ne laisse pas de place à l'approximatif. L'intuition du mathématicien lui donne les idées d'une preuve, mais rendre cette preuve rigoureuse est parfois un travail de longue haleine.  $\text{LEAN}$  répond à ce problème puisque son objectif est de répondre à une simple question à chaque étape : "Ai-je le droit de faire ceci?".

L'application des règles de logique entre plusieurs énoncés, basés sur un système cohérent d'axiomes, permet à  $\text{LEAN}$  de vérifier la véracité des preuves mais également de les rendre plus cohérente : l'ordre des arguments, la nécessité ou non de telle hypothèse, etc...

$\text{LEAN}$  est donc un outil à la résolution des problèmes mathématiques, mais il ne remplacera jamais l'esprit acéré du mathématicien devant son tableau noir.

Depuis quelques années, la communauté  $\text{LEAN}$ , constituée de chercheurs en mathématiques et de passionnés, œuvre dans le but de constituer une librairie suffisante pour que l'utilisation de  $\text{LEAN}$  devienne accessible à tous niveaux.

La librairie *mathlib* rassemble un grand nombre de définitions, lemmes et théorèmes avec leurs preuves, sur des domaines très variés allant de la topologie à l'algèbre des modules passant par la théorie de la mesure. L'objectif est de formaliser les résultats "undergraduate", c'est-à-dire jusqu'au niveau Master principalement.

Toutes ces preuves existantes peuvent être utilisées directement dans les preuves que nous faisons, afin de ne pas avoir à re-démontrer certains résultats préliminaires.

Ce rapport présentera, dans un premier temps, l'utilisation basique du logiciel en prenant exemple sur des résultats simples, puis quelques résultats d'analyse complexe du premier semestre de Master, dont nous détailleront les preuves.

## 2 LEAN et ses tactiques

Le langage `L $\exists\forall$ N` a un grand intérêt lors de l'écriture des preuves. En effet, à chaque étape, `L $\exists\forall$ N` affiche le *contexte local* et l'*objectif*.

Le *contexte local* réunit toutes les données existantes à l'instant  $T$ , que ce soit les variables introduites dans l'énoncé du théorème ou dans la preuve, mais aussi les hypothèses sur ces variables. L'*objectif* est, au début, l'énoncé à prouver. À chaque ligne, celui-ci s'actualise afin de donner exactement les éléments qu'il reste à prouver.

Cette configuration permet de réfléchir à la preuve directement sur le logiciel, mais aussi de comprendre plus facilement les erreurs que nous aurions pu faire.

Pour résoudre un objectif, c'est-à-dire faire une preuve, il nous faudra utiliser des tactiques et des lemmes. Les tactiques sont des outils permettant le raisonnement mathématique pur, par exemple remplacer dans une équation une variable par une autre dont on a prouvé qu'elles sont égales. Les lemmes sont des énoncés que l'on a prouvé précédemment ou qui se trouvent dans la librairie. Nous y ferons appel, en les adaptant au contexte local.

La combinaison de ces deux éléments permet de réaliser les preuves, de la plus simple à la plus complexe. Parfois, pour résoudre un objectif complexe, nous serons amenés à créer des lemmes intermédiaires, ou encore à avoir plusieurs objectifs dans la même preuve.

Un autre élément important se trouve dans la syntaxe des preuves. On trouvera toujours une virgule après une commande, elle est indispensable pour que `L $\exists\forall$ N` interprète la tactique.

### 2.A – Les tactiques de résolution – exemple

Voici un énoncé simple que nous allons étudier dans un premier temps.

#### Code lean : Inégalités de réels

```
1 example (a b c : ℝ) (hc : 0 ≤ c) (hab : a ≤ b) : a*c ≤ b*c :=
2 begin
3   rw ← sub_nonneg,
4   have h_fact : b*c - a*c = (b - a)*c,
5   { ring },
6   rw h_fact,
7   apply mul_nonneg,
8   { rw sub_nonneg,
9     exact hab },
10  { exact hc },
11 end
```

À la première lecture, tout ceci doit vous sembler quelque peu incompréhensible. Prenons les choses une par une.

Tout d'abord, la ligne 1 : c'est l'énoncé. Le mot clef `example` annonce à `L $\exists\forall$ N` un énoncé que l'on ne souhaite pas garder en mémoire pour pouvoir le réutiliser, à l'instar de `lemma` que nous verrons plus tard.

La syntaxe pour ce mot clef est la suivante : `example (variables) (hypothèses) : résultat :=`.

Ici, on déclare 3 variables  $a$ ,  $b$ ,  $c$  qui sont des réels, ou plutôt qui sont "de type" réel. On donne ensuite deux hypothèses :

- Une appelée `hc` qui dit que  $0$  est plus petit que  $c$  ;
- l'autre, `hab`, donne  $a$  plus petit ou égal à  $b$ .

Enfin, on annonce le résultat que l'on souhaite prouver :  $ac \leq bc$ . Ce résultat est, a priori, très simple. Il va cependant nécessiter quelques tactiques pour le prouver.

La preuve du résultat se trouve après les symboles `:=` et entre les mots `begin` et `end`. Si nous plaçons notre curseur juste après le `begin`, voici ce que `L3VN` affiche :

```
1 goal
2 a b c : ℝ
3 hc: 0 ≤ c
4 hab: a ≤ b
5 ⊢ a * c ≤ b * c
```

On observe sur les lignes 2,3 et 4 le *contexte local* avec les variables et les hypothèses données dans l'énoncé, puis en ligne 5, après le symbole  $\vdash$ , l'*objectif* en cours.

Commençons maintenant la preuve. Pour cela nous aurons besoin de quelques lemmes existant déjà dans la librairie `mathlib` :

```
1 sub_nonneg : 0 ≤ b - a ↔ a ≤ b,
2 mul_nonneg : (0 ≤ a → 0 ≤ b) → 0 ≤ a * b
```

La première étape va être de transformer l'objectif  $ac \leq bc$  en  $0 \leq bc - ac$ . Nous allons donc utiliser le lemme `sub_nonneg` et la tactique `rewrite`

Tactique	<b>Rewrite – rw</b> La tactique <code>rw</code> réécrit l'objectif en cours à l'aide d'une égalité. On peut la traduire par "On remplace".	Si l'objectif est $a = c$ et une hypothèse $h : a = b$ , alors écrire <code>rw h</code> donne l'objectif $b = c$ .
----------	---	--

La tactique `rw` fonctionne également avec les équivalences : si  $h : P \leftrightarrow Q$  est une hypothèse, alors `rw h` transforme  $P \rightarrow R$  en  $Q \rightarrow R$ .

Une variante est nécessaire dans notre cas, puisque l'on veut réécrire l'implication inverse. Pour cela, il suffit d'ajouter `←` après le `rw`. On écrit donc `rw ← sub_nonneg`, et voici ce que `L3VN` nous dit :

```
1 a b c : ℝ
2 hc : 0 ≤ c
3 hab : a ≤ b
4 ⊢ 0 ≤ b * c - a * c
```

Passons à la ligne 4. Nous voulons maintenant factoriser le membre de droite de notre égalité. Pour cela, nous allons avoir recours à la tactique `have` :

Tactique	<b>Have</b> Elle a pour effet de créer une nouvelle hypothèse, sous condition d'en donner une preuve dans le contexte local. La syntaxe est la suivante : <code>have 'nom' : 'hypothèse'</code> .	Écrire <code>have h : a = b</code> donne deux objectifs : <ul style="list-style-type: none"> <li>– Prouver que <math>a = b</math> avec le contexte local ;</li> <li>– l'objectif initial dont le contexte local est enrichi avec l'hypothèse <math>h</math>.</li> </ul>
----------	---	---

Ici, on va donc créer l'hypothèse nommée `h_fact` qui donne la factorisation : c'est la ligne 4.

Il nous faut ensuite donner la preuve de cette factorisation. Bien heureusement, nous n'avons pas à reprendre toutes les mathématiques de base, il nous suffit d'utiliser le raccourci `ring`. Celui-ci va

tout simplement résoudre les objectifs avec les règles de calculs propres aux anneaux (comme le nom l'indique), comme dans notre exemple.

On peut finalement réécrire cette nouvelle hypothèse dans l'objectif par un simple `rw h_fact`, et on obtient l'objectif suivant :

```
1 ⊢ 0 ≤ (b - a) * c
```

Nous avons maintenant quelque chose de la même forme que la conclusion du lemme `mul_nonneg` et nous voudrions pouvoir revenir à ses arguments, "prendre la flèche dans l'autre sens". Pour cela, nous allons introduire la tactique `apply`.

Tactique	<p style="text-align: center;"><b>Apply</b></p> <p>Apply cherche une ressemblance entre l'objectif et la conclusion du lemme. Il change ensuite l'objectif en cours en demandant les arguments du lemme en question.</p>	<p>Si on a <code>lemme_1 : A → B</code> et un objectif de la forme <code>B</code>, écrire <code>apply lemme_1</code> change l'objectif par <code>A</code>.</p>
----------	--	--

Ainsi, appliquons le lemme `mul_nonneg` à l'objectif, nous obtenons les objectifs suivants :

```
1 2 goals
2
3 abc: ℝ
4 hc: 0 ≤ c
5 hab: a ≤ b
6 h_fact: b * c - a * c = (b - a) * c
7 ⊢ 0 ≤ b - a
8
9 abc: ℝ
10 hc: 0 ≤ c
11 hab: a ≤ b
12 h_fact: b * c - a * c = (b - a) * c
13 ⊢ 0 ≤ c
```

Concernant le premier objectif, celui-ci ressemble fortement à l'hypothèse `hab`. Pour avoir une ressemblance exacte, il nous faut appliquer notre lemme `sub_nonneg` avec la tactique `rw` : en effet, ici on veut remplacer notre objectif  $0 \leq b - a$  par  $a \leq b$ . On peut ensuite conclure à l'aide de la tactique `exact`

Tactique	<p style="text-align: center;"><b>Exact</b></p> <p>Comme son nom l'indique, cette tactique agit uniquement quand l'objectif est exactement le même qu'une hypothèse ou qu'un lemme connu. Elle permet de conclure la démonstration.</p>	<p>Si on a une hypothèse <code>h : a = b</code> et que l'objectif est <code>a = b</code>, alors <code>exact h</code> permet de conclure.</p>
----------	---	--

Ici, on conclut donc le premier objectif.

Pour le second, c'est exactement l'hypothèse `hc`, donc on conclut avec la tactique `exact`.

Nous avons donc réussi à prouver que multiplier par un nombre positif ne change pas le sens d'une inégalité. C'est évidemment un résultat très simple que nous n'aurons plus jamais besoin de démontrer, puisqu'il est dans la librairie sous le nom de `mul_mono_nonneg`.

On peut d'ailleurs écrire comme seule démonstration :

```
1 exact mul_mono_nonneg hc hab,
```

Il s'agit du lemme correspondant que l'on applique aux hypothèses du contexte local.

Il reste nombre d'autres tactiques que l'on peut utiliser, nous ne les énuméreront pas toutes ici. Voici simplement les quelques autres tactiques dont nous aurons besoin par la suite.

Pour la première, `L3V3N` connaît déjà nombre de choses en mathématiques, et pour faire appel à ces connaissances, on peut demander à `L3V3N` de simplifier l'objectif :

<b>Tactique</b>	<p><b>Simplify – simp</b></p> <p><code>L<sub>3</sub>V<sub>3</sub>N</code> va simplifier les énoncés de l'objectif à l'aide des lemmes qu'il connaît. On peut également lui demander de simplifier certaines définitions propres à notre code ou encore de s'aider des hypothèses.</p>	<p>Il existe plusieurs syntaxe pour <code>simp</code> :</p> <ul style="list-style-type: none"> <li>– <code>simp</code> utilise certains lemmes de la librairie,</li> <li>– <code>simp[blabla]</code> utilise certains lemmes et les hypothèses <code>blabla</code> qui lui sont données.</li> </ul>
-----------------	---	---

C'est une tactique bien utile pour simplifier les objectifs sans avoir à rechercher tous les lemmes correspondants dans la librairie.

## 2.B – Les tactiques de travail

Parlons maintenant d'une tactique plus complexe mais pourtant bien utile pour les preuves longues : `refine`. Nous l'utiliseront à de nombreuses reprises dans la suite, prenons donc le temps de l'expliquer ici.

<b>Tactique</b>	<p><b>Refine</b></p> <p>On utilise cette tactique pour séparer un objectif complexe en autant d'objectifs qu'il y a d'arguments. Il suffit d'indiquer autant de <code>_</code> que d'arguments et chaque argument devient un objectif.</p>	<p>Par exemple, si on veut une application linéaire de <math>\mathbb{R}</math> dans <math>\mathbb{R}</math>, c'est-à-dire un objectif <math>\mathbb{R} \rightarrow_l \mathbb{R}</math>, écrire <code>refine ⟨_,_,_⟩</code> permet d'avoir 3 objectifs (3 <code>_</code>) qui corresponde à donner l'application, prouver son additivité et prouver son homogénéité (cf plus loin).</p>
-----------------	--	--

Nous verrons l'utilité de cette tactique plus loin.

Lorsque nous travaillons sur une preuve, qui n'est donc pas finie, `L3V3N` annonce à chaque ligne des erreurs liées à la fin de la preuve. Afin d'éviter cette surcharge d'informations, nous pouvons utiliser le mot magique `sorry`. Lors de la recherche et de la rédaction d'une preuve, on va par habitude écrire `sorry` à la fin, où sur chaque élément de preuve de sorte que `L3V3N` accepte notre lemme et que l'on puisse l'utiliser par la suite, sans avoir encore fini la preuve.

Enfin, terminons par une tactique qui permet de chercher s'il existe un résultat semblable à l'objectif dans la librairie, qui permettrait de conclure : `library_search`.

## 3 L'analyse complexe par LEAN

Afin de rendre les connaissances acquises via les tutoriels d'utilisation de `L3V3N`, nous allons coder les résultats et les preuves du cours d'analyse complexe de M1. En raison de la longueur du travail nécessaire pour une seule preuve, nous n'iront pas jusqu'au bout du cours.

### 3.A – Cadre

Tout d'abord, posons le cadre. Heureusement, la librairie `mathlib` possède déjà une version de  $\mathbb{C}$ . Les nombres complexes sont présentés comme des paires de réels, la partie réelle et la partie imaginaire.

Autrement dit, le code `z : ℂ` signifie `x y : ℝ, z = (x, y)`.

Cette définition va nous être bien utile par la suite. Pour y faire appel dans notre fichier `L3VN`, il nous suffit d'écrire

```
1 import analysis.complex.basic
```

Une des conséquences évidentes est donc que  $\mathbb{C}$  est en bijection avec  $\mathbb{R} \times \mathbb{R}$ . Ceci est donné par la fonction `complex.equiv_real_prod`. Par exemple, si  $z : \mathbb{C}$ , alors `complex.equiv_real_prod z` est un élément de  $\mathbb{R} \times \mathbb{R}$ . En revanche, si  $x : \mathbb{R} \times \mathbb{R}$ , alors `complex.equiv_real_prod.symm x` est un nombre complexe. En effet, `complex.equiv_real_prod` est une fonction de  $\mathbb{C}$  dans  $\mathbb{R} \times \mathbb{R}$  qui est bijective, dont on peut donc prendre la réciproque via la commande `.symm`.

Une autre propriété est que cette fonction qui lie  $\mathbb{C}$  et  $\mathbb{R} \times \mathbb{R}$  est linéaire continue. Pour cela on utilise la fonction `complex.equiv_real_prod` dont la linéarité et la continuité sont prouvées.

Pour continuer, nous aurons besoin des fonctions holomorphes, qui sont en réalité des fonctions dérivables de  $\mathbb{C}$ . Il se trouve que la dérivation a été définie, tout comme la différentiabilité (qui sont fortement liées l'une à l'autre...) dans la librairie. Ces notions, comme toutes les autres dans la librairie, sont définies de la manière la plus générale possible, ce qui signifie que nous pouvons utiliser la dérivation dans  $\mathbb{C}$  par la même commande que celle dans  $\mathbb{R}$ , ce qui est très pratique.

Afin d'utiliser ces définitions, importons le bon fichier :

```
1 import analysis.calculus.deriv
```

Pour finir, ajoutons le code

```
1 noncomputable theory
```

afin d'éviter la plupart des problèmes qui pourrait survenir.

### 3.B – Cauchy-Riemann, étape 1 – énoncé

Commençons par ce premier résultat, les équations de Cauchy-Riemann. En revanche, nous n'allons pas pouvoir le prouver sous la forme la plus connue, c'est-à-dire par les dérivées partielles, puisque celles-ci n'existent pas directement dans la librairie. Nous allons passer par une version quelque peu différente. Voici le premier résultat à prouver.

#### Holomorphe $\implies$ différentiable sur $\mathbb{R}^2$

Soit  $f : \mathbb{C} \rightarrow \mathbb{C}$  une fonction holomorphe ( $\mathbb{C}$ -dérivable) en  $z \in \mathbb{C}$ . Alors  $f$ , en tant qu'application de  $\mathbb{R}^2$  dans  $\mathbb{R}^2$  est différentiable en  $z \in \mathbb{R}^2$  de différentielle la multiplication par  $f'(z)$ .

La première étape est de réussir à écrire un énoncé accepté par `L3VN`. On a ici 3 variables : la fonction  $f$ , le point  $z$  et le point  $f'(z)$  que l'on notera  $f'$ . On travaille ici en un unique point, il n'y a pas de nécessité de considérer la fonction dérivée  $f' : \mathbb{C} \rightarrow \mathbb{C}$ .

On a ensuite une hypothèse de départ :  $f$  est  $\mathbb{C}$ -dérivable en  $z$  de dérivée  $f'$  (encore une fois,  $f'$  est un point qui est par définition la dérivée de  $f$  en  $z$ ). Le théorème qui dit que  $f$  est  $\mathbb{C}$ -dérivable en un point porte le nom de `has_deriv_at`. On peut donc commencer par écrire :

```
1 lemma cauchy_riemann_step_1 {f : ℂ → ℂ} {z : ℂ} (f' : ℂ) (hf : has_deriv_at
  f f' z) :
```

Parlons maintenant de l'énoncé à prouver. On veut voir  $f$  comme une fonction de  $\mathbb{R}^2$ . Pour cela, nous allons écrire `realify f`, puis nous définirons la fonction `realify` plus loin. Le théorème qui dit qu'une fonction est différentiable est `has_fderiv_at`. On peut appeler `multiply` la fonction de multiplication que nous définirons plus loin. On a alors :

```
1 lemma cauchy_riemann_step_1 {f : ℂ → ℂ} {z : ℂ} (f' : ℂ) (hf : has_deriv_at
  f f' z) :
2 has_fderiv_at (realify f) (multiply f') (complex.equiv_real_prod z) :=
```



Sous réserve de définir `realify` et `multiply`, cet énoncé tient la route.

On notera d'ailleurs la syntaxe de la commande `lemma` : en premier le nom que l'on donne afin d'y faire référence plus tard. Ensuite, entre parenthèse ou accolades, les variables et les hypothèses, puis `:`, l'énoncé et on termine par `:=`.

### 3.C – Un premier essai

Nous avons tout d'abord défini `realify f` comme étant la composition de `f` avec les fonctions qui vont de  $\mathbb{C}$  dans  $\mathbb{R}^2$  et réciproquement. Avec la fonction `f` en argument, nous obtenons le code suivant :

```
1 def realify (f :  $\mathbb{C} \rightarrow \mathbb{C}$ ) : ( $\mathbb{R} \times \mathbb{R} \rightarrow \mathbb{R} \times \mathbb{R}$ ) := equiv_real_prod.to_fun  $\circ$  f  $\circ$ 
   equiv_real_prod.inv_fun
```

On notera l'utilisation ici des attributs `to_fun` et `inv_fun` qui permettent l'utilisation d'un sens ou de l'autre d'une bijection.

Une autre nouveauté dans cette ligne de code est la commande `def`. Comme son nom l'indique, elle permet de définir un objet, notamment des fonctions. On écrit donc `def` puis le nom de la fonction, on ajoute `:` et on écrit le type de l'objet, ici une fonction de  $\mathbb{R}^2$  dans  $\mathbb{R}^2$ . Enfin, après `:=`, on donne la définition.

Nous avons maintenant notre fonction qui transforme une fonction sur  $\mathbb{C}$  en une fonction sur  $\mathbb{R}^2$ . Il nous faut maintenant notre multiplication. Nous allons la définir comme une application linéaire et continue de  $\mathbb{R}^2$  dans  $\mathbb{R}^2$ , ce qui nous donne :

```
1 def multiply (z :  $\mathbb{C}$ ) : ( $\mathbb{R} \times \mathbb{R} \rightarrow_{L[\mathbb{R}]} \mathbb{R} \times \mathbb{R}$ ) :=
```

Notons que le caractère linéaire et continue est donné par la syntaxe  $\rightarrow_{L[\mathbb{R}]}$ . Nous n'allons pas pouvoir ici juste dire qu'il s'agit de la multiplication, puisqu'il nous faudra aussi prouver la linéarité et la continuité. Pour observer tout ceci, nous allons utiliser `refine`.

```
1 def multiply (z :  $\mathbb{C}$ ) : ( $\mathbb{R} \times \mathbb{R} \rightarrow_{L[\mathbb{R}]} \mathbb{R} \times \mathbb{R}$ ) := by {
2   refine  $\langle \_, \_ \rangle$ ,
```

On a deux `_`, donc deux choses à donner à `L $\exists$ VN` :

```
1 2 goals
2 z :  $\mathbb{C}$ 
3  $\vdash \mathbb{R} \times \mathbb{R} \rightarrow_{L[\mathbb{R}]} \mathbb{R} \times \mathbb{R}$ 
4 z :  $\mathbb{C}$ 
5  $\vdash$  auto_param (continuous ?m_1.to_fun) (name.mk_string "continuity"
   (name.mk_string "interactive" (name.mk_string "tactic" name.anonymous)))
```

Premièrement, `L $\exists$ VN` attends une fonction  $\mathbb{R}$ -linéaire de  $\mathbb{R}^2$  dans  $\mathbb{R}^2$ . Puis, le second objectif est la continuité de cette fonction (dit dans des termes bien complexes).

On utilise une nouvelle fois `refine` pour avoir le détail du premier objectif :

```
1 def multiply (z :  $\mathbb{C}$ ) : ( $\mathbb{R} \times \mathbb{R} \rightarrow_{L[\mathbb{R}]} \mathbb{R} \times \mathbb{R}$ ) := by {
2   refine  $\langle \_, \_ \rangle$ ,
3   { refine  $\langle \_, \_, \_ \rangle$ ,
```

L'utilisation des accolades permet de se concentrer sur un seul objectif à la fois. Voici ce que nous dit `L $\exists$ VN` :

```
1 3 goals
2 z :  $\mathbb{C}$ 
3  $\vdash \mathbb{R} \times \mathbb{R} \rightarrow \mathbb{R} \times \mathbb{R}$ 
4 z :  $\mathbb{C}$ 
5  $\vdash \forall (x\ y : \mathbb{R} \times \mathbb{R}), ?m_1 (x + y) = ?m_1\ x + ?m_1\ y$ 
6 z :  $\mathbb{C}$ 
7  $\vdash \forall (r : \mathbb{R}) (x : \mathbb{R} \times \mathbb{R}), ?m_1 (r \cdot x) = \uparrow(\text{ring\_hom.id } \mathbb{R})\ r \cdot ?m_1\ x$ 
```



Voici donc ce qu'est une application linéaire : c'est la donnée d'une application, la preuve de son additivité et celle de son homogénéité.

On donne donc l'application : il s'agit de la réalification de la multiplication dans  $\mathbb{C}$  par  $z$ , puis les preuves qui sont assez élémentaires puisque  $L\mathbb{E}VN$  sait le faire :

```
1 def multiply (z : ℂ) : (ℝ × ℝ →L[ℝ] ℝ × ℝ) := by {
2   refine ⟨_,_⟩,
3   { refine ⟨_,_,_⟩,
4     { exact realify (λ w, z * w) },
5     { intros, simp [realify], split; ring },
6     { intros, simp [realify], split; ring } },
```

Il nous reste maintenant à prouver la continuité de cette application. Cela peut paraître élémentaire, mais essayons de le faire avec  $L\mathbb{E}VN$ . Tout d'abord, on utilise `simp` afin de comprendre l'objectif, on obtient :

```
1 1 goal
2 z: ℂ
3 ⊢ continuous (realify (has_mul.mul z))
```

Ici, `has_mul.mul` est le nom donné par  $L\mathbb{E}VN$  pour la fonction que nous avons défini juste avant. Afin de prouver la continuité, on va utiliser la caractérisation de Lipschitz, donnée par `lipshitz_with`, et prouver que notre application est lipschitzienne avec une constante de Lipschitz  $K = 2\|z\|$ . Pour cela, on utilise la commande `suffices` qui traduit "il suffit de" afin d'amener un nouvel objectif qui permettras de conclure. Voici le code :

```
1 def multiplication (z : ℂ) : (ℝ × ℝ →L[ℝ] ℝ × ℝ) := by {
2   refine ⟨_,_⟩,
3   { refine ⟨_,_,_⟩,
4     { exact realify (λ w, z * w) },
5     { intros, simp [realify], split, ring, ring },
6     { intros, simp [realify], split; ring } },
7   { simp,
8     suffices : lipshitz_with (nnnorm z * 2) (realify (has_mul.mul z)),
```

et le résultat donné par  $L\mathbb{E}VN$  :

```
1 2 goals
2 z: ℂ
3 this: lipshitz_with (||z||_+ * 2) (realify (has_mul.mul z))
4 ⊢ continuous (realify (has_mul.mul z))
5 z: ℂ
6 ⊢ lipshitz_with (||z||_+ * 2) (realify (has_mul.mul z))
```

Il nous faut donc, dans un premier temps, prouver que savoir la fonction Lipschitzienne permet de conclure à sa continuité. Pour cela, on utilise un lemme de la librairie, `lipshitz_with.continuous`, qui dit exactement ce qu'il nous faut. On écrit alors :

```
1 def multiplication (z : ℂ) : (ℝ × ℝ →L[ℝ] ℝ × ℝ) := by {
2   refine ⟨_,_⟩,
3   { refine ⟨_,_,_⟩,
4     { exact realify (λ w, z * w) },
5     { intros, simp [realify], split, ring, ring },
6     { intros, simp [realify], split; ring } },
7   { simp,
8     suffices : lipshitz_with (nnnorm z * 2) (realify (has_mul.mul z)),
9     exact lipshitz_with.continuous this,
```

Il reste maintenant à prouver que notre fonction est Lipschitzienne. C'est malheureusement quelque chose qui n'est pas simple. Après de nombreuses recherches, voici le code nécessaire pour prouver cette caractérisation lipschitzienne :

### Code lean : La multiplication est lipschitzienne

```

1 def rabs : ℝ → ℝ := abs
2
3 lemma l1 (z w : ℂ) : (realify (has_mul.mul z) (equiv_real_prod w)) =
  equiv_real_prod.to_fun (z * w) := rfl
4
5 lemma l2 (z1 z2 : ℂ) : equiv_real_prod (z1 - z2) = equiv_real_prod z1 -
  equiv_real_prod z2 := by { rfl }
6 lemma l3 (z1 z2 : ℂ) : equiv_real_prod (z1 - z2) = equiv_real_prod.to_fun
  z1 - equiv_real_prod.to_fun z2 := by { rfl }
7
8 lemma l6 (z : ℂ) : nnnorm (equiv_real_prod z) ≤ nnnorm z := by {
9   rw prod.nnnorm_def, simp [nnnorm, norm], split,
10  apply abs_re_le_abs, apply abs_im_le_abs
11 }
12
13 lemma l7 (z : ℂ) : nnnorm z ≤ 2 * nnnorm (equiv_real_prod z) := by {
14   simp [nnnorm, norm], rw ← subtype.coe_le_coe, simp,
15   transitivity rabs z.re + rabs z.im,
16   exact complex.abs_le_abs_re_add_abs_im z,
17   rw two_mul, apply add_le_add, apply le_max_left, apply le_max_right,
18 }
19
20 lemma l5 {z xy : ℂ} : ||equiv_real_prod (z * xy)||+ ≤ nnnorm z * 2 * ||
  equiv_real_prod xy||+ :=
21 begin
22   transitivity nnnorm (z * xy), apply l6,
23   simp_rw mul_assoc,
24   simp,
25   apply mul_le_mul le_rfl,
26   apply l7, apply zero_le, apply zero_le,
27 end
28
29 lemma l8 (z : ℂ) : lipschitz_with (nnnorm z * 2) (realify (has_mul.mul z))
  :=
30 begin
31   rw [lipschitz_with], intros x y,
32   set x' := equiv_real_prod.inv_fun x with h1,
33   set y' := equiv_real_prod.inv_fun y with h1,
34   have : x = equiv_real_prod x' := by { simp [x'] }, rw this, rw l1,
35   have : y = equiv_real_prod y' := by { simp [y'] }, rw this, rw l1,
36   rw [edist_eq_coe_nnnorm_sub, edist_eq_coe_nnnorm_sub, ←l2, ←l3],
37   rw ← mul_sub,
38   set xy := x' - y' with hxy, rw ←hxy,
39   rw ← ennreal.coe_mul,

```

```

40  apply ennreal.coe_le_coe.mpr,
41  exact l5
42  end

```

Et ainsi, on peut conclure la définition de multiplication par un simple exact l8 z.

Pour cette preuve, il a nécessité un grand nombre de lemmes intermédiaires lors de la recherche. Il est possible de réduire grandement la rédaction de ces lemmes grâce aux raccourcis d'écriture de  $L\exists\forall N$ .

Nous ne l'avons pas fait ici car, en tentant la preuve de notre `cauchy_riemann_step_1`, nous nous sommes aperçu que cette définition de multiplication n'est pas la plus simple à manipuler. Nous l'avons donc laisser de côté, ou plutôt, nous avons adapté nos énoncés.

### 3.D – Un simple changement

Pour commencer et afin de simplifier l'utilisation, re-définissons les fonctions de  $\mathbb{C}$  dans  $\mathbb{R}^2$  :

```

1  def C_to_R2 :  $\mathbb{C} \rightarrow L[\mathbb{R}] \mathbb{R} \times \mathbb{R} := \text{complex.equiv\_real\_prod}_1$  -- l'application de
    $\mathbb{C}$  dans  $\mathbb{R}^2$ 
2  def R2_to_C :  $\mathbb{R} \times \mathbb{R} \rightarrow L[\mathbb{R}] \mathbb{C} := \text{complex.equiv\_real\_prod}_1.\text{symm}$  -- sa reciproque

```

Ce sont les fonctions que nous avons vu depuis le début avec la particularité qu'elles sont directement fournies avec la preuve de leur continuité et de leur linéarité (la seule différence est le petit  $_1$  après le nom).

Re-définissons maintenant le `realify` avec ces fonctions.

```

1  def realify (f :  $\mathbb{C} \rightarrow \mathbb{C}$ ) :  $\mathbb{R} \times \mathbb{R} \rightarrow \mathbb{R} \times \mathbb{R} := \text{C\_to\_R2} \circ f \circ \text{R2\_to\_C}$ 

```

Au final, nous avons donné la même définition à `realify`, à la différence près des petites fonctions `C_to_R2` et `R2_to_C` qui sont légèrement différentes des précédentes. Cette petite différence va nous être bien utile par la suite.

On peut d'ailleurs voir que la définition de la multiplication du paragraphe précédent est maintenant beaucoup plus aisée avec ces deux fonctions `C_to_R2` et `R2_to_C`.

#### Code lean : La multiplication par un complexe dans $\mathbb{R}^2$ est $\mathbb{R}$ -linéaire et continue

```

1  def mul_exe (z :  $\mathbb{C}$ ) : ( $\mathbb{R} \times \mathbb{R} \rightarrow L[\mathbb{R}] \mathbb{R} \times \mathbb{R}$ ) := by {
2    refine ⟨_,_⟩, -- on recommence les refine, comme avant
3    { refine ⟨_,_,_⟩,
4      -- on veut la multiplication par z, mais de  $\mathbb{R}^2$  dans  $\mathbb{R}^2$ 
5      { exact realify (λ w, w * z) },
6      -- LEAN prouve l'additivité avec la tactique ring
7      { intros, simp [realify], ring },
8      -- on continue avec l'homogénéité, encore une fois avec ring,
9      { intros, simp [realify, C_to_R2], split ; ring },
10   },
11   -- on simplifie et on applique la règle de continuité sur la composition
12   simp [realify], apply continuous.comp,
13   -- C_to_R2 est continue
14   { exact C_to_R2.continuous },
15   -- encore la continuité de la composition
16   apply continuous.comp,
17   -- la multiplication à droite est continue

```

```

18 { exact continuous_mul_right z },
19 -- R2_to_C est continue
20 { exact R2_to_C.continuous },
21 }

```

En revanche, ce n'est toujours pas cette multiplication que nous allons utiliser car elle ne permet pas de conclure sur notre énoncé `cauchy_riemann_step_1`.

Nous allons plutôt utiliser la multiplication par un complexe comme une application de  $\mathbb{C}$  dans  $\mathbb{C}$  qui serait continue et  $\mathbb{R}$ -linéaire. Pour cela, définissons un premier élément : la multiplication par un complexe de  $\mathbb{C}$  dans  $\mathbb{C}$  est  $\mathbb{C}$ -linéaire et continue. Le code est sensiblement le même que celui ci-dessus :

### Code lean : La multiplication par un complexe dans $\mathbb{C}$ est $\mathbb{C}$ -linéaire et continue

```

1 def multiplication (w : ℂ) : ℂ →L[ℂ] ℂ :=
2   begin
3     refine ⟨_,_⟩, -- on demande à LEAN de generer les objectifs de la
4       definition
5     {
6       refine ⟨_,_,_⟩, -- encore une fois
7       exact λ z, w * z, -- LEAN veut une application de ℂ dans ℂ, on lui
8         donne la multiplication par w ∈ ℂ
9       exact mul_add w, -- on trouve ensuite la propriete de linearite (on
10        utilise notamment library_search)
11       intros, simp, ring, -- on termine avec des tactiques simples
12     },
13     simp, -- on demande à LEAN de simplifier pour y voir clair
14     exact continuous_mul_left w, -- encore un library_search pour trouver la
15       propriete dans mathlib
16   end

```

On notera que certaines étapes se résolvent de façon beaucoup plus simple, avec l'aide de la librairie et de la tactique `library_search`.

Le code ci-dessus représente la méthode de recherche des preuves. Dans la librairie notamment, les résultats ne sont pas présentés ainsi, mais en version raccourcie. Ici, nous pouvons raccourcir la preuve en utilisant les symboles `⟨⟩` comme ceux utilisés avec `refine`. Plus précisément, on va remplacer les `_` utilisés avec `refine` par les éléments de preuve correspondants :

### Code lean : La définition précédente – en version condensée

```

1 def multiply (w : ℂ) : ℂ →L[ℂ] ℂ :=
2   ⟨⟨λ z, w * z, mul_add w, by { intros, simp, ring }⟩, continuous_mul_left w⟩

```

La suite consiste à prouver que cette multiplication est également  $\mathbb{R}$ -linéaire, ou plutôt de définir une multiplication de  $\mathbb{C}$  dans  $\mathbb{C}$  qui est  $\mathbb{R}$ -linéaire à partir de `multiply`. Pour cela, nous allons utiliser une propriété qui implique la continuité et la linéarité sur le corps des scalaires à partir de celles sur l'espace vectoriel, qui s'appelle `continuous_linear_map.restrict_scalars`. Autrement dit, on écrit :

## Code lean : La multiplication restreinte au corps des scalaires

```

1 def real_multiply (f' :  $\mathbb{C}$ ) :  $\mathbb{C} \rightarrow \mathbb{L}[\mathbb{R}] \mathbb{C} :=$ 
2   continuous_linear_map.restrict_scalars  $\mathbb{R}$  (multiply f')

```

Nous allons maintenant pouvoir revenir à notre `cauchy_riemann_step_1`. Mais, avant de le prouver, nous allons devoir adapter l'énoncé avec les nouvelles définitions que nous avons prises. En effet, l'argument de dérivée que nous avions était simplement `multiply f'` qui devait être une application de  $\mathbb{R}^2$  dans  $\mathbb{R}^2$ ,  $\mathbb{R}$ -linéaire et continue. Ici, nous n'avons plus qu'une multiplication dans  $\mathbb{C}$  qui est  $\mathbb{R}$ -linéaire et continue.

Nous allons donc devoir transformer cette application. À première vue, nous pourrions simplement appliquer `realify` à notre fonction. Or, la définition de `realify` n'assure pas la continuité de la composition de fonctions, même si celle-ci est vraie. Nous allons donc définir une version linéaire et continue de `realify` :

```

1 def realifyl (f :  $\mathbb{C} \rightarrow \mathbb{L}[\mathbb{R}] \mathbb{C}$ ) :  $\mathbb{R} \times \mathbb{R} \rightarrow \mathbb{L}[\mathbb{R}] \mathbb{R} \times \mathbb{R} := \text{C\_to\_R2} \circ \text{f} \circ \text{R2\_to\_C}$ 

```

et enfin donner notre énoncé :

## Code lean : Nouvel énoncé pour 3.B

```

1 lemma cauchy_riemann_step_1 {f :  $\mathbb{C} \rightarrow \mathbb{C}$ } {z :  $\mathbb{C}$ } (f' :  $\mathbb{C}$ ) (hf :
   has_deriv_at f f' z) :
2   has_fderiv_at (realify f) (realifyl (real_multiply f')) (C_to_R2 z) :=

```

Il nous reste alors la preuve, qui se déroule de façon conventionnelle, avec pour principal outil le théorème `has_fderiv_at.comp` qui n'est rien d'autre que le théorème de dérivation en chaîne. Voici la preuve :

## Code lean : Preuve de la première étape de Cauchy-Riemann

```

1 lemma cauchy_riemann_step_1 {f :  $\mathbb{C} \rightarrow \mathbb{C}$ } {z :  $\mathbb{C}$ } (f' :  $\mathbb{C}$ ) (hf :
   has_deriv_at f f' z) :
2   has_fderiv_at (realify f) (realifyl (real_multiply f')) (C_to_R2 z) :=
3   begin
4     -- On donne la preuve que R2_to_C est l'inverse à gauche de C_to_R2
5     have zz : function.left_inverse R2_to_C C_to_R2 :=
6       complex.equiv_real_prod.left_inv,
7     -- on applique la règle de dérivation en chaîne
8     apply has_fderiv_at.comp,
9     -- la preuve que C_to_R2 soit édiffrentiable de édiffrentielle elle-même
10    est :
11    { apply C_to_R2.has_fderiv_at,
12      -- on applique encore une fois la règle de la chaîne
13      { apply has_fderiv_at.comp,
14        -- on demande à LEAN de comparer l'hypothèse hf restreinte à  $\mathbb{R}$  avec le
15        goal
16        -- LEAN donne alors à prouver les édiffrences
17        { convert has_fderiv_at.restrict_scalars  $\mathbb{R}$  hf.has_fderiv_at,
18          -- on simplifie le goal : on éveloppe real_multiply, puis multiply
19          { simp [real_multiply, multiply,

```

```

    continuous_linear_map.restrict_scalars],
17   -- deux applications sont égales si elles sont égales en tout point
18   apply linear_map.ext, intro x, simp, apply mul_comm
19 }, -- il reste maintenant à appliquer l'hypothèse zz que nous avons
    émontre
20 { apply zz }, },
21 -- et voici la preuve que R2_to_C est édiffrentiable de édiffrentielle
    elle-même
22 { apply R2_to_C.has_fderiv_at, }, },
23 end

```

Enfin, la dernière étape, qui n'est pas obligatoire, mais qui correspond au travail mené pour la librairie, est de rendre la preuve plus courte. Elle sera uniquement plus courte en nombre de lignes, car le contenu et la méthode seront les mêmes. Voici ce que l'on obtient :

#### Code lean : Preuve précédente – version condensée

```

1 lemma cauchy_riemann_step_1 {f : ℂ → ℂ} {z : ℂ} (f' : ℂ) (hf :
    has_deriv_at f f' z) :
2   has_fderiv_at (realify f) (realify (real_multiply f')) (C_to_R2 z) :=
3   begin
4     refine C_to_R2.has_fderiv_at.comp _ (has_fderiv_at.comp _ _
        R2_to_C.has_fderiv_at),
5     have zz : function.left_inverse R2_to_C C_to_R2 :=
        complex.equiv_real_prod.left_inv,
6     rw zz z,
7     convert has_fderiv_at.restrict_scalars ℝ hf.has_fderiv_at,
8     simp [real_multiply, multiply, continuous_linear_map.restrict_scalars],
9     apply linear_map.ext, intro z, simp, apply mul_comm,
10  end

```

### 3.E – Cauchy-Riemann step 2

L'objectif est d'exprimer clairement les relations de Cauchy-Riemann avec les dérivées partielles. Pour cela, nous allons devoir parler de la matrice de la notre application de multiplication.

Commençons par définir la forme générale d'une matrice de multiplication. Pour définir une matrice, on définit un objet mathématique du type `matrix`. On doit donner aussi la taille. Pour cela, on va utiliser les types de la forme `fin n` : c'est le sous-type de  $\mathbb{N}$  composé des entiers strictement inférieurs à  $n$ . Comme on travaille sur  $\mathbb{R}^2$ , on veut des matrices de tailles  $2 \times 2$  à coefficients dans  $\mathbb{R}$ . Ensuite, pour définir une matrice, la syntaxe est sensiblement la même que la plupart des langages de programmation. Voici la définition :

#### Code lean : Matrice de multiplication

```

1 def mulmatrix (a b : ℝ) : matrix (fin 2) (fin 2) ℝ :=
2   ![![a, -b],
3     ![b, a]]

```

Nous allons donc discuter de la matrice de la multiplication par  $f'$ , c'est-à-dire la matrice avec partie réelle et imaginaire de  $f'$  :

$$F = \begin{bmatrix} \operatorname{Re}(f') & -\operatorname{Im}(f') \\ \operatorname{Im}(f') & \operatorname{Re}(f') \end{bmatrix}$$

Cette matrice correspond donc, d'après notre définition, à `mulmatrix (f'.re) (f'.im)`. On veut donc montrer que notre application `real_multiply f'` dans sa version réalifiée, s'exprime par la matrice  $F$  ci-dessus. Pour cela, on va plutôt montrer que l'application linéaire qui vient de la matrice  $F$  est la même que `real_multiply f'`.

Les outils dont nous auront besoin sont les suivants :

- `matrix.to_lin'` est un lemme qui donne l'équivalence entre une matrice et une application linéaire. Ainsi, `matrix.to_lin' (mulmatrix (f'.re) (f'.im))` est une application linéaire de  $(\text{fin } 2 \rightarrow \mathbb{R})$  dans  $(\text{fin } 2 \rightarrow \mathbb{R})$ .
- `fin_two_arrow_equiv` qui est l'équivalence entre le type  $(\text{fin } 2 \rightarrow \alpha)$  et  $\alpha \times \alpha$ . Alors, `fin_two_arrow_equiv  $\mathbb{R}$`  est l'équivalence entre  $(\text{fin } 2 \rightarrow \mathbb{R})$  et  $\mathbb{R} \times \mathbb{R}$ .

Nous voulons donc prouver cet énoncé :

```
1 lemma cauchy_riemann_step_2 (f' : ℂ) :
2   (fin_two_arrow_equiv ℝ) ∘ matrix.to_lin' (mulmatrix (f'.re) (f'.im)) ∘
3   (fin_two_arrow_equiv ℝ).symm
4   = realify₁ (real_multiply f') :=
```

La preuve est finalement assez simple, elle paraît même être triviale. Du côté de `LEAN`, nous utilisons tout d'abord le fait que nos deux fonctions sont égales si elles sont égales sur tout élément de  $\mathbb{R} \times \mathbb{R}$ , puis plusieurs simplifications successives et enfin une résolution évidente avec `ring`. Voici la preuve :

#### Code lean : Preuve Cauchy-Riemann step 2

```
1 lemma cauchy_riemann_step_2 (f' : ℂ) :
2   (fin_two_arrow_equiv ℝ) ∘ matrix.to_lin' (mulmatrix (f'.re) (f'.im)) ∘
3   (fin_two_arrow_equiv ℝ).symm
4   = realify₁ (real_multiply f') :=
5   begin
6     funext, -- deux fonctions sont les êmmes si elles sont les êmmes sur tout
7       élément de l'ensemble
8     simp [realify₁, C_to_R2, R2_to_C, mulmatrix, real_multiply, multiply],
9     split ; ring,
10  end
```